# Example Technology Exercise 12
# James Jones

Load the "vect" utility file so we have access to cross products.
Define the norm, projection, cross product, and unitvector

```
(%i1) load(vect)$
      norm(u):=sqrt(u.u)$
      proj(u,v):=(u.v)/(v.v)*v$
      cross(u,v):=express(u~v)$
      unit(u):=u/norm(u)$
```

Load the draw function to get the 3D graph. I expect you to use
Winplot on the tech project, but this is included as an example.

```
(%i6) load(draw)$
```

## 1 Problem 12.4.77

Define problem

```
(%i7) r:[4*sin(t),4*cos(t),2*t]$
      t0:%pi/3$
```

```
T = r'/||r'||, B = r'xr''/||r'xr''||, N = BxT
```

```
(%i9) r1:diff(r,t)$
      r2:diff(r,t,2)$
      T:trigsimp(unit(r1));
      B:trigsimp(unit(cross(r1,r2)));
      N:cross(B,T);
```

$$(\%o11)\ [\frac{2\cos(t)}{\sqrt{5}},-\frac{2\sin(t)}{\sqrt{5}},\frac{1}{\sqrt{5}}]$$

$$(\%o12)\ [\frac{\cos(t)}{\sqrt{5}},-\frac{\sin(t)}{\sqrt{5}},-\frac{2}{\sqrt{5}}]$$

$$(\%o13)\ [-\sin(t),-\cos(t),0]$$

Now evaluate them at the point t0

```
(%i14) T0:subst(t=t0,T);
       N0:subst(t=t0,N);
       B0:subst(t=t0,B);
```

$$(\%o14)\ \left[\frac{1}{\sqrt{5}},-\frac{\sqrt{3}}{\sqrt{5}},\frac{1}{\sqrt{5}}\right]$$

$$(\%o15)\ \left[-\frac{\sqrt{3}}{2},-\frac{1}{2},0\right]$$

$$(\%o16)\ \left[\frac{1}{2\sqrt{5}},-\frac{\sqrt{3}}{2\sqrt{5}},-\frac{2}{\sqrt{5}}\right]$$

To sketch the vectors in winplot, we need
the initial point and terminal point

The initial point is r(t0)

```
(%i17) r0:subst(t=t0,r)$
```

Now for the terminal points of the other vectors
Put this into decimal form to make easier to enter into Winplot

```
(%i18) float(r0);
       float(r0+T0);
       float(r0+N0);
       float(r0+B0);
```

$$(\%o18)\ [3.464101615137754,2.0,2.094395102393195]$$

$$(\%o19)\ [3.911315210637712,1.225403330758517,2.541608697893153]$$

$$(\%o20)\ [2.598076211353316,1.5,2.094395102393195]$$

$$(\%o21)\ [3.687708412887734,1.612701665379258,1.19996791139328]$$

This is not required for the technology project, it's included
here for instructional purposes

```
Notes about the following commands

The dimensions=[600,400] command may not be implemented in older
versions like the one installed on Richland's computers.

To get the view, use draw3d() intead of wxdraw3d() and then use
the mouse to drag the image to the view you want.  The view will
be at the bottom of the screen.  If you do use draw() instead of
wxdraw(), then you must close the graph window before Maxima will
respond or continue processing.

The 1.1 times the vector on the labels is to put the label past
the vector so you can see it.

The xyplane should match the minimum z or it will add extra space
at the bottom to get down to z=0.
```

```
(%i22) Tvec:vector(r0,T0)$
       Nvec:vector(r0,N0)$
       Bvec:vector(r0,B0)$
       curve:parametric(r[1],r[2],r[3],t,t0-0.2,t0+0.2)$
       point:points([r0])$

       Tlabel:label(append(["T"],r0+1.1*T0))$
       Nlabel:label(append(["N"],r0+1.1*N0))$
       Blabel:label(append(["B"],r0+1.1*B0))$
```

```
(%i30) set_draw_defaults(
          xrange=[2,4],yrange=[1,3],zrange=[1,3],
          line_width=2,grid=true,xyplane=1,
          point_size=1.5,point_type=filled_circle,
          dimensions=[400,400],
          view=[34,57]
       )$
```

```
(%i31) wxdraw3d(
         color=red,Tvec,Tlabel,
         color=dark_green,Nvec,Nlabel,
         color=orange,Bvec,Blabel,
         color=blue,curve,point
       )$
```

(%t31)




## 2 Problem 12.5.12

Define the curve

```
(%i32) r:[2*sin(t),5*t,2*cos(t)]$
       t0:0$
       t1:%pi$
```

Find ||r'|| and integrate

```
(%i35) trigsimp(norm(diff(r,t)))$
       integrate(%,t,t0,t1);
```

(%o36) $\sqrt{29}\,\pi$

# 3 Problem 12.5.43

Define problem

```
(%i37) r:[t,t^2,t^3/4];
       P:[2,4,2];
```

$$(\%o37)\ [t,t^2,\frac{t^3}{4}]$$

$$(\%o38)\ [2,4,2]$$

Find value of t0.  Pick the x coordinate since it's linear.

```
(%i39) linsolve(r[1]=P[1],t);
       t0:rhs(%[1]);
```

$$(\%o39)\ [t=2]$$

$$(\%o40)\ 2$$

Find the derivatives at t0

```
(%i41) diff(r,t)$
       r1:subst(t=t0,%);
       diff(r,t,2)$
       r2:subst(t=t0,%);
```

$$(\%o42)\ [1,4,3]$$

$$(\%o44)\ [0,2,3]$$

Find the curvature

```
(%i45) K:norm(cross(r1,r2))/norm(r1)^3;
```

$$(\%o45)\ \frac{7}{26^{3/2}}$$